
Introduction

There are many applications that need editors such as HotDraw [1] for manipulating graphical objects on the screen. Unfortunately, such frameworks tend to be relatively complex and bulky. It would be nice to have a simple pluggable framework built-in to the Smalltalk environment. Currently, there isn’t so we either have to deal with the complexities of these general purpose facilities or build our own. In our case, we wanted to build a simple editor that could support the manipulation of points, lines, and regions. Such an editor might be used for network layout, simple top-view home architecture building, or game development to name a few. In our case, the editor that we will discuss is an amalgamation of classes that were worked on by a number of people including Wayne Beaton, Anthony Lander, and Jon Hylands. 



Our inspiration came from a recent desire to look at the DEU editor [2] (the Doom Editor Utility written in C) which was designed to manipulate DOOM game files. Not having time to browse it online, we simply printed it so that we could take the listings home for light reading over the weekend. It was a bit of a shock to find eight inches of paper when we finally went to the print room to pick it up. Jon Hylands then suggested we look at the start of such an editor that he wrote in Smalltalk (but never quite finished). It’s considerably less code!!! We reworked his implementation to make it more generic; i.e., to separate out the editing framework from the editor itself. Currently, the framework supports the ability to create, move, and edit the three graphical objects mentioned above but doesn’t support resizing or zooming (see Figure 1 for the editor in action; items 1, 2, and 3 respectively illustrate a selected point, a selected line, and a selected area).



Figure 1

The Design

The editor was designed to execute in ParcPlace/Digitalk Visual Smalltalk Enterprise 3.1. Seven classes are provided as shown in Figure 2; two classes (ScrollingPane and DrawingPane) provide the generic framework, and five classes (RegionEditor along with RegionEditorObject and its three subclasses) provide the editor functionality. ScrollingPane is a replacement for GraphPane that doesn’t exhibit clipping region problems when scrolling to the border regions. As the name implies, we probably don’t want to use it as a substitute for GraphPane although it would work. DrawingPane is a graph pane that support drawing and manipulating collections of objects. 



Figure 2

The Generic Framework Classes

The ScrollingPane class, unlike the GraphPane class, is designed to control scrolling in absolute rather than relative terms. To understand how to use it, we need to distinguish between pane coordinates (where 0@0 is the top-left corner of the pane) and scrolling (or canvas) coordinates where the top-left corner is the origin of a very large area; we’ll refer to the dimensions of this area as the scroll rectangle. In general, the pane only shows a portion of the canvas(scrolling is used to move the visible portion around. As we scroll, then, a particular point in a picture has pane coordinates that keep changing whereas the canvas coordinates are constant. All the protocol (see below) that we provide deals with canvas coordinates. Drawing with the pane’s pen at specified canvas coordinates will result in drawing at canvas locations (as expected). Scrolling permits a user to move the pane’s top-left corner to cover the entire interior of the scrolling rectangle. As shown in Figure 3, this implies that it is possible to view a portion the size of the pane which lies to the right and bottom of the scrolling rectangle.



Figure 3



	ScrollingPane Protocol

		scrollRectangle: canvasRectangle

		scrollRectangle

		scrollExtent 

		scrollPosition: canvasPoint

		scrollPosition

		scrollToMakeRectangleVisible: cr

		mouseLocation 

		whileButton1DownDo: aBlock

		mapCanvasToPane: point/rect

		mapPaneToCanvas: point/rect

	Note: cr is short for canvasRectangle.



We aren’t going to discuss (or list) the ScrollingPane methods since they are very similar in spirit to the GraphPane methods. But there is one small difference. Rather than signal a single #display event when redraw is required, we also signal a #display: event which provides the rectangle (in canvas coordinates) that represents the area to be redrawn. If, for instance, we move another window over the top-right portion of the pane, only that portion of the pane needs to be redrawn when the overlapping window is moved away. For what we are about to do, however, we will not actually take advantage of this feature. 



The DrawingPane class inherits all the scrolling functionality from ScrollingPane but has the extra task of managing a collection of objects. In general, it doesn’t know what the objects are for, so it can’t draw them by itself. What it needs to do is inform the application when interesting events occur. In the current design, the following interesting events are triggered.



	#needsMenu (inherited)

	#objects:near: (objects aPoint)

	#objects:movedBy: (objects aPoint)

	#objectsDeleted: (objects)

	#objectsDisplaySelected:unselected: 

		(objects1 object2)

	#singleClickSelected: (objects)

	#doubleClickSelected: (objects)

	#controlClickAt: (aPoint)

	Note: Points are in canvas coordinates.



The first event, #needsMenu, requires the application to set the pane’s menu via “aPane setMenu: aMenu”(this is a standard built-in event. The second event, #objects:near:, is expected to return a subset of the objects provided(those which are near aPoint. As a consequence, applications must have at most one handler for this event (only the last handler’s answer is returned). Event #objects:movedBy: is triggered after the user has dragged a number of objects to a new destination. The objects moved along with the amount by which they moved (the delta) is provided as parameter. Event #objectsDeleted: is triggered after the currently selected objects have been removed from the pane as a consequence of the delete (or backspace) key having been hit. Normally, the application will do nothing (the thank you very much response) but sometimes, it may have to re-insert some deleted objects (perhaps some objects are not allowed to be deleted for some obscure reason). The #objectsDisplaySelected:unselected: event tell the application to draw some objects (the first collection consists of the selected objects and the second, the unselected objects). The #singleClickSelected: event is triggered when a user selects a collection of objects. The drawing pane supports clicking on individual objects to select them or using the shift key to add an unselected object to the existing list or remove a previously selected object from the list. The  #singleClickSelected: event is generated whenever the list of selected objects changes. The  #doubleClickSelected: is generated whenever the user double-clicks the left mouse button. Control click also triggers an event, #controlClickAt:, which we needed because the editor supports the addition of points in the middle of a line with a control click on the line.



Normally, an application taps into the events at the time its panes are created. So, for example, we might specify a handler as follows:



	aPane 

		when: #singleClickSelected: 

		send: #handleSingleClickSelected: 

		to: anApplication



Method handleSingleClickSelected: in the application will then be supplied with the collection of selected objects. If we don’t want to receive the provided parameters (for whatever reason), we can supply our own at construction time but these values will not change during execution. So, for example, we could tap into the event as follows:



	aPane 

		when: #singleClickSelected: 

		send: #handleSingleClickSelected: 

		to: anApplication

		with: aPane



Now, method handleSingleClickSelected: will be supplied with “aPane” instead of the collection of selected objects. If the pane is a drawing pane (which it should be for what we might want to do), the selected objects can be obtained by directly querying the pane as in the following:



	aPane selectedObjects



As designers of the events, we do try to anticipate what information will be needed by the application (and supply it as parameters) but we can’t anticipate all contingencies. Sometimes, the application will have to communicate directly with the pane.



Class DrawingPane (see Listing 1) is responsible for maintaining the graphical objects, keeping track of which ones are selected, and converting lower level events (such as mouse clicking) to higher level events (those described above). Let’s trace what happens, for example, when the user shift-clicks on an unselected object(the intent is to add the object to the current list of objects. Ultimately, we will have to trigger a #singleClickSelect: event to indicate that the set of selected objects has changed. The drama starts when the user shift-clicks with the left button. The operating system sends the message #wmLbuttondown:with: to the drawing pane (the method is inherited from ScrollingPane which converts the pane coordinates of the mouse to canvas coordinates, records it in mouseLocation (in case a user subsequently queries the pane), and in turn sends the message #button1Down: with the converted point. This method (see Listing 1) distinguishes three cases: the shift key is down, the control key is down, and the normal case. Since the shift key is down, in this case, message shiftSelect: does the work. It begins by asking the application for all objects near the click point. Each of these objects is either added to the current list of selected objects or removed from the list depending on whether or not it was already there. Subsequently, we trigger the #singleClickSelected: event (which typically is ignored by the application) and then send the invalidate message. This particular message causes a wmPaint:with: message which ultimately results in a display message that forces the application to redraw its objects (via event #objectsDisplaySelected:unselected:).



Listing 1



Another interesting aspect is the double buffering that drawing panes provide. When a user asks the pane for its pen, he is provided with the pen of the bitmap instead (maintained in instance variable buffer). This pen is specially offset so that the user can use it to draw in canvas coordinates. It is the pane’s responsibility to copy the bitmap to the pane as a consequence of an invalidate (or display) message. This makes it a little tricky when you read the DrawingPane methods; “self pen” provides the bitmap’s pen whereas “super pen” provides the pane’s pen. From the user’s perspective, there is only one pen (he doesn’t need to know where he is drawing; as far as he is concerned, it’s the pane but we know better).

The Region Editor Classes

The region editor contains a single drawing pane and serves as an illustration of the utility of the DrawingPane facilities. Originally, we started with a version that implemented everything in the editor because it was using a pane equivalent to a graph pane and we reworked it to use a drawing pane. We didn’t, however, try to change its basic design. We can add a region (via the right mouse button popup menu), add an additional vertex to an existing line (via control selection on a line), select any of the objects (areas, lines, or vertices with a higher priority for the more basic parts; area selection, however, is inaccurate because it is based on the bounding box), and delete regions (we don’t support deletion of parts yet). 



One complication with the existing design is that an area object, when added to the pane, also causes its parts (both the lines and the vertices) to be added as independent objects. Thus it is possible to shift-click select a number of regions, independent lines, and also individual vertices and also move them around (a nice facility for game building). The drawing pane, however, doesn’t know about the parts interactions. Deleting an area, for example, should also cause the lines and vertices to disappear. To support the deletion of areas, we have to additionally remove objects that were not originally selected; i.e., the area parts. Since we don’t currently support deleting lines and vertices, we have to put such objects back if they are not part of an area that is to be deleted. We contemplated a redesign of the region editor objects to simplify some of this complexity but we’ll have to leave that for you if you’re interested.



Let’s consider one of the event handlers, the #objects:movedBy: handler shown below. Assuming that objects might be areas, lines, and vertices, some vertices might be supplied multiple times; i.e., individually, as part of a line, and also as part of an area object. We must ensure that such a vertices are adjusted only once. So we first gather them in a set (by asking for each object’s lowest component; the set removes duplicates) and then we add the move delta (aPoint) to the remaining vertices.



	objects: objects movedBy: aPoint

		| vertices |

		vertices := objects 

			inject: Set new into: [:set :object |

				set 

					addAll: object 

						lowestComponents; 

					yourself].

		vertices do: [:vertex |

			vertex position: vertex position + 

				aPoint]



Another interesting method is the #objects:near: event handler. Sometimes, the drawing pane wants to know which of the selected objects are near the point (when the user is about to drag the selected objects); at other times, which of all the pane objects are near the point (when new selections are being attempted). The editor responds by returning only vertices (if there any). Otherwise, it returns lines (if any). As a last resort, it returns areas. Different kinds of objects respond differently to message isNearPoint:.  



	objects: objects near: aPoint

		"Return the most primitive type

		of object first."

		| list |

		list := objects select: [:object | 

			object isVertex and: [

			object isNearPoint: aPoint]].

		list isEmpty ifFalse: [^list].

		list := objects select: [:object | 

			object isLine and: [

			object isNearPoint: aPoint]].

		list isEmpty ifFalse: [^list].

		^objects select: [:object | 

			object isArea and: [

			object isNearPoint: aPoint]]



See Listing 2 for more details about the region editor. Because of space limitations, we couldn’t provide listings for the region editor object and its subclasses. But those objects are relatively obvious anyway.

Conclusions

The drawing pane provides pluggable facilities for applications that need to provide simple facilities for manipulating graphical objects. The event-oriented architecture makes it very easy to plug-in arbitrary objects since the pane sends absolutely no messages to the objects directly. 

Listing 1  The DrawingPane class.



class:									DrawingPane

superclass: 						ScrollingPane 

instanceVariableNames:	objects selectedObjects buffer 

poolDictionaries:				OperatingSystemConstants

											VirtualKeyConstants



class methods



events supported

constructEventsTriggered

	"Private - answer the set of events that instances of the receiver can trigger."

	"DrawingPane initializeEventsTriggered"

	^super constructEventsTriggered

		add: #controlClickAt:;

		add: #objects:near: ;

		add: #objects:movedBy:;

		add: #objectsDeleted:;

		add: #objectsDisplaySelected:unselected:;

		add: #singleClickSelected:;

		add: #doubleClickSelected:;

		yourself



instance methods



accessing

selectedObjects: aCollection

	selectedObjects := aCollection

selectedObjects

	^selectedObjects

objects

	^objects

objects: aCollection

	objects := aCollection

pen

	^buffer pen setViewportOrg: super pen getViewportOrg



object adding

addObject: anObject

	^self addObject: anObject refresh: true

addObjects: aCollection

	^self addObjects: aCollection refresh: true

addObject: anObject refresh: aBoolean

	self objects add: anObject.

	aBoolean ifTrue: [self invalidate]

addObjects: aCollection refresh: aBoolean

	aCollection do: [:object | self objects add: object].

	aBoolean ifTrue: [self invalidate]



object deleting

deleteObject: anObject

deleteObject: anObject refresh: aBoolean

deleteObjects: aCollection

	Similar in structure to add methods.

deleteObjects: aCollection refresh: aBoolean

	aCollection do: [:object |

		self objects remove: object ifAbsent: [].

		self selectedObjects remove: object ifAbsent: []].

	aBoolean ifTrue: [self invalidate]



object selecting

selectObject: anObject

selectObject: anObject refresh: aBoolean

selectObjects: aCollection

	Similar in structure to add methods.

selectObjects: aCollection refresh: aBoolean

	self selectedObjects: aCollection.

	aBoolean ifTrue: [self invalidate]



private-initializing

initialize

	super initialize.

	self objects: OrderedCollection new.

	self selectedObjects: OrderedCollection new.

defaultCursor

	^CursorManager crossHair



private-buffer

destroyBuffer

	buffer isNil ifFalse: [buffer release]

createBuffer

	self destroyBuffer.

	buffer := Bitmap screenExtent: 

		self visibleRectangle extent.

	self pen

		foreColor: self foreColor;

		backColor: self backColor



private-displaying

displayIn: aRectangle

	“Triggered by wmPaint:with: caused by invalidate.”

	super displayIn: aRectangle.

	self display

displayBuffer

	super pen

		copyBitmap: buffer

		from: self visibleRectangle

		to: self visibleRectangle

displayObjects

	| unselected |

	unselected := self objects reject: [:object |

		self selectedObjects includes: object].

	self triggerEvent: #objectsDisplaySelected:unselected:

		with: self selectedObjects with: unselected

display

	self pen

		fill: self visibleRectangle

		color: self backColor.

	self

		displayObjects;

		displayBuffer



private-validating

validate

	super validate.

	self createBuffer



private-resizing

sizeChanged: anExtent

	super sizeChanged: anExtent.

	self createBuffer



private-selecting

moveFrom: aPoint

	"Inform owner of each move."

	| start |

	start := aPoint. self invalidate.

	self whileButton1DownDo: [:location |

		self triggerEvent: #objects:movedBy: 

			with: self selectedObjects with: location - start.

		start := self mouseLocation.

		self display].

normalSelect: aPoint

	"Selecting a preselected object begins a move; otherwise, a new list of selections is computed. Inform owner afterwards."

	| list |

	list := self triggerEvent: #objects:near: 

		with: self selectedObjects with: aPoint.

	list isEmpty ifFalse: [^self moveFrom: aPoint].

	list := self triggerEvent: #objects:near: 

		with: self objects with: aPoint.

	self selectObjects: list.

	self triggerEvent: #singleClickSelected: with: list.

	self invalidate.

	self moveFrom: aPoint

controlSelect: aPoint

	"User-defined activity required. Informs owner."

	self triggerEvent: #controlClickAt: with: aPoint.

	self invalidate

shiftSelect: aPoint

	"Automatically adds/removes from selected objects. Informs owner afterwards."

	| list |

	list := self triggerEvent: #objects:near: 

		with: self objects with: aPoint.

	list do: [:object |

			(self selectedObjects includes: object)

				ifTrue: [self selectedObjects remove: object]

				ifFalse: [self selectedObjects add: object]].

	self triggerEvent: #singleClickSelected: 

		with: self selectedObjects.

	self invalidate



private-closing

close

	super close.

	buffer release



private-window events

button1Down: aPoint

	"Handle 3 modifier possibilities in canvas coordinates."

	super button1Down: aPoint.

	self isShiftKeyDown 

		ifTrue: [^self shiftSelect: aPoint].

	self isControlKeyDown 

		ifTrue: [^self controlSelect: aPoint].

	^self normalSelect: aPoint

button1DoubleClick: aPoint

	"In canvas coordinates."

	super button1DoubleClick: aPoint.

	self triggerEvent: #doubleClickSelected: 

		with: self selectedObjects.

	self invalidate

button2Up: aPoint

	"In canvas coordinates."

	super button2Up: aPoint.

	self triggerEvent: #needsMenu: with: self selectedObjects

keyboardInput: aKeyboardInputEvent

	"Private - keyboard input was received.  Process backspace, tab, carriage return."

	| virtualKey |

	( virtualKey := aKeyboardInputEvent virtualKey ) notNil 	ifTrue: [

			(virtualKey == BackspaceKey or: [

				virtualKey == DeleteKey])

					ifTrue: [self deleteSelectedObjects]].

deleteSelectedObjects

	| candidates |

	candidates := self selectedObjects shallowCopy.

	self deleteObjects: candidates refresh: false.

	self triggerEvent: #objectsDeleted: with: candidates; invalidate

 





Listing 2  The region editor.



class:									RegionEditor

superclass: 						ApplicationCoordinator

instanceVariableNames:	drawingPane



class methods



building

buildView: aView forModel: anApplication

	"Create the structure of the window."

	^anApplication buildView: aView



instance methods



event handlers

objects: objects movedBy: aPoint

	"The selected objects have been moved by aPoint. Change their position."

	| vertices |

	vertices := objects inject: Set new into: [:set :object |

		set addAll: object lowestComponents; yourself].

	vertices do: [:vertex |

		vertex  position: vertex position + aPoint]



objects: objects near: aPoint

	" Return the most primitive type of object first."

	| list |

	list := objects select: [:object | object isVertex and: 

		[object isNearPoint: aPoint]].

	list isEmpty ifFalse: [^list].

	list := objects select: [:object | object isLine and: 

		[object isNearPoint: aPoint]].

	list isEmpty ifFalse: [^list].

	^objects select: [:object | object isArea and: 

		[object isNearPoint: aPoint]].



objectsDisplaySelected: selected unselected: unselected

	"Display the objects."

	unselected do: [:object | 

		object displayWith: drawingPane pen].

	selected do: [:object | 

		object displaySelectedWith: drawingPane pen].



objectMenu: aPane

	"Set the menu of the drawingPane"

	| menu |

	menu := Menu labels: 

		'&Add Region\&Delete Region\&Inspect' withCrs

		actions: #(addRegion deleteRegion inspect).

	menu owner: self.

	aPane setMenu: menu



objectsDeleted: objects

	"We will only handle deletion of areas."

	| areas components deleteSet addSet |

	areas := objects select: [:object | object isArea].

	components := areas inject: Set new into: [:set :object |

		set addAll: object components; yourself].

	deleteSet := components select: [:object |

		drawingPane objects includes: object].

	addSet := objects reject: [:object |

		components includes: object].

	drawingPane deleteObjects: deleteSet refresh: false.

	drawingPane addObjects: addSet refresh: false



objectControlClickIn: aPane

	"Insert new vertex in the middle of the line that is nearby."

	| aPoint areas startVertex endVertex newVertex newLine |

	aPoint := aPane mouseLocation.

	aPane objects do: [:object |

		(object isLine and:[object isNearPoint: aPoint]) ifTrue: [

			areas := object areas.

			startVertex := object vertices first.

			endVertex := object vertices last.

			newVertex := RegionEditorVertex from: aPoint.

			newLine := RegionEditorLine 

				from: newVertex 

				to: endVertex.

			object replace: endVertex with: newVertex.

			areas do: [:eachArea | eachArea 

				add: newVertex between: startVertex and: endVertex.

				eachArea addLine: newLine].

			aPane

				addObject: newVertex refresh: false;

				addObject: newLine refresh: false;

				selectObject: newVertex.

			^aPane moveFrom: aPoint]]



menu handlers

deleteRegion

	"Delete the selected region from the drawing."

	drawingPane deleteSelectedObjects.



addRegion

	"Add a region to the drawing."

	| aPoint vertexPoints area newObjects |

	aPoint := drawingPane mouseLocation.

	vertexPoints := OrderedCollection

		with: (aPoint leftAndUp: 10 @ 10)

		with: (aPoint rightAndUp: 10 @ 10)

		with: (aPoint rightAndDown: 10 @ 10)

		with: (aPoint leftAndDown: 10 @ 10).

	area := RegionEditorArea from: vertexPoints.

	newObjects := area components.

	newObjects do: [:object |

		drawingPane addObject: object refresh: false].

	drawingPane selectObject: area



private-building

buildView: aView

	"Build the view of the application."

	aView

		label: 'Region Editor';

		noSmalltalkMenuBar;

		addSubpane: ((drawingPane := DrawingPane new)

			owner: self;

			when: #controlClickAt: 

				send: #objectControlClickIn: to: self 

				with: drawingPane;

			when: #objects:movedBy: 

				send: #objects:movedBy: to: self;

			when: #objectsDeleted: 

				send: #objectsDeleted: to: self;

			when: #objects:near: 

				send: #objects:near: to: self;

			when: #objectsDisplaySelected:unselected: 

				send: #objectsDisplaySelected:unselected: to: self;

			when: #needsMenu 

				send: #objectMenu: to: self with: drawingPane).

	drawingPane foreColor: Color black

Figure 1  The region editor.

Figure 2  The class hierarchy structure.

Figure 3  When the scrolling location is restricted to the scroll rectangle, the visible area includes the dark areas to the right and below the scroll rectangle.

