class:									RegionEditor


superclass: 						ApplicationCoordinator


instanceVariableNames:	drawingPane





class methods





building





buildView: aView forModel: anApplication


	"Create the structure of the window."


	^anApplication buildView: aView





instance methods





event handlers





objects: objects movedBy: aPoint


	"The selected objects have been moved by aPoint. Change their position."


	| vertices |


	vertices := objects inject: Set new into: [:set :object |


		set addAll: object lowestComponents; yourself].


	vertices do: [:vertex |


		vertex  position: vertex position + aPoint]





objects: objects near: aPoint


	" Return the most primitive type of object first."


	| list |


	list := objects select: [:object | object isVertex and: 


		[object isNearPoint: aPoint]].


	list isEmpty ifFalse: [^list].


	list := objects select: [:object | object isLine and: 


		[object isNearPoint: aPoint]].


	list isEmpty ifFalse: [^list].


	^objects select: [:object | object isArea and: 


		[object isNearPoint: aPoint]].





objectsDisplaySelected: selected unselected: unselected


	"Display the objects."


	unselected do: [:object | 


		object displayWith: drawingPane pen].


	selected do: [:object | 


		object displaySelectedWith: drawingPane pen].





objectMenu: aPane


	"Set the menu of the drawingPane"


	| menu |


	menu := Menu labels: 


		'&Add Region\&Delete Region\&Inspect' withCrs


		actions: #(addRegion deleteRegion inspect).


	menu owner: self.


	aPane setMenu: menu





objectsDeleted: objects


	"We will only handle deletion of areas."


	| areas components deleteSet addSet |


	areas := objects select: [:object | object isArea].


	components := areas inject: Set new into: [:set :object |


		set addAll: object components; yourself].


	deleteSet := components select: [:object |


		drawingPane objects includes: object].


	addSet := objects reject: [:object |


		components includes: object].


	drawingPane deleteObjects: deleteSet refresh: false.


	drawingPane addObjects: addSet refresh: false





objectControlClickIn: aPane


	"Insert new vertex in the middle of the line that is nearby."


	| aPoint areas startVertex endVertex newVertex newLine |


	aPoint := aPane mouseLocation.


	aPane objects do: [:object |


		(object isLine and:[object isNearPoint: aPoint]) ifTrue: [


			areas := object areas.


			startVertex := object vertices first.


			endVertex := object vertices last.


			newVertex := RegionEditorVertex from: aPoint.


			newLine := RegionEditorLine 


				from: newVertex 


				to: endVertex.


			object replace: endVertex with: newVertex.


			areas do: [:eachArea | eachArea 


				add: newVertex between: startVertex and: endVertex.


				eachArea addLine: newLine].


			aPane


				addObject: newVertex refresh: false;


				addObject: newLine refresh: false;


				selectObject: newVertex.


			^aPane moveFrom: aPoint]]





menu handlers





deleteRegion


	"Delete the selected region from the drawing."


	drawingPane deleteSelectedObjects.





addRegion


	"Add a region to the drawing."


	| aPoint vertexPoints area newObjects |


	aPoint := drawingPane mouseLocation.


	vertexPoints := OrderedCollection


		with: (aPoint leftAndUp: 10 @ 10)


		with: (aPoint rightAndUp: 10 @ 10)


		with: (aPoint rightAndDown: 10 @ 10)


		with: (aPoint leftAndDown: 10 @ 10).


	area := RegionEditorArea from: vertexPoints.


	newObjects := area components.


	newObjects do: [:object |


		drawingPane addObject: object refresh: false].


	drawingPane selectObject: area





private-building





buildView: aView


	"Build the view of the application."


	aView


		label: 'Region Editor';


		noSmalltalkMenuBar;


		addSubpane: ((drawingPane := DrawingPane new)


			owner: self;


			when: #controlClickAt: 


				send: #objectControlClickIn: 


				to: self 


				with: drawingPane;


			when: #objects:movedBy: 


				send: #objects:movedBy:  


				to: self;


			when: #objectsDeleted: 


				send: #objectsDeleted: 


				to: self;


			when: #objects:near: 


				send: #objects:near:  


				to: self;


			when: #objectsDisplaySelected:unselected: 


				send: #objectsDisplaySelected:unselected: 


				to: self;


			when: #needsMenu 


				send: #objectMenu: 


				to: self 


				with: drawingPane).


	drawingPane foreColor: Color black











class:									RegionEditorObject


superclass: 						Object 


instanceVariableNames:	position 


poolDictionaries:				OperatingSystemConstants





class methods





instance creation


  


new


	^super new initialize





instance methods





initializing





initialize


	"Do nothing"





accessing





recalculatePosition


	"Do nothing"





position


	^position


  


boundingBox


	^position extentFromLeftTop: self extent





extent


	^self implementedBySubclass





testing


 


isNearPoint: aPoint


	^self boundingBox containsPoint: aPoint


 


isVertex


	^false


 


isLine


	^false


   


isArea


	^false





displaying





displaySelectedWith: aPen


	| oldColor |


	oldColor := aPen foreColor.


	aPen foreColor: Color darkRed.


	self displayWith: aPen.


	aPen foreColor: oldColor





printing


 


printOn: aStream


	aStream


		nextPutAll: self printOnDescription;


		nextPutAll: ' ('.


	self printInternalOn: aStream.


	aStream nextPut: $)


  


printOnDescription


	^self class name











class:									RegionEditorArea


superclass: 						RegionEditorObject 


instanceVariableNames:	vertices lines


poolDictionaries:				OperatingSystemConstants





class methods





instance creation





from: aCollection


	^self new from: aCollection





instance methods





accessing





vertexPoints


	^vertices collect: [:each | each position]





vertices


	^vertices


  


add: aVertex between: startVertex and: endVertex


	| startIndex endIndex |


	aVertex addArea: self.


	startIndex := vertices indexOf: startVertex.


	endIndex := vertices indexOf: endVertex.


	startIndex < endIndex | (endIndex = 1)


		ifTrue: [vertices add: aVertex afterIndex: startIndex]


		ifFalse: [vertices add: aVertex afterIndex: endIndex]


 


recalculatePosition


	(vertices isNil or: [vertices isEmpty])


		ifTrue: [^self].


	position := self vertexPoints


		inject: vertices first position


		into: [:sum :each |


			sum leftMostAndHighest: each]


   


replace: aVertex with: anotherVertex


	| index |


	index := vertices indexOf: aVertex.


	vertices at: index put: anotherVertex.


	anotherVertex addArea: self.


	self recalculatePosition


 


addLine: aLine


	(self includesLine: aLine) ifFalse: [


		lines add: aLine.


		aLine addArea: self]


   


lowestComponents


	^vertices


  


extent


	^(self vertexPoints


		inject: vertices first position


		into: [:sum :each |


			sum rightMostAndLowest: each]) - position


 


lines


	^lines





components


	^(WriteStream on: (Array new: 20))


		nextPut: self;


		nextPutAll: lines;


		nextPutAll: vertices;


		contents


  


removeLine: aLine


	lines remove: aLine ifAbsent: []





testing


  


includesLine: aLine


	lines do: [:each |


		(each isSameAs: aLine)


			ifTrue: [^true]].


	^false


   


isArea


	^true





displaying





displaySelectedWith: aPen


	aPen


		setLineWidth: 1;


		setHatchBrush: HsFdiagonal color: aPen foreColor.


	aPen polygonFilled: self vertexPoints.


	aPen


		setBrushStyle: BsSolid color: aPen foreColor hatch: nil





displayWith: aPen


	aPen


		setLineWidth: 1;


		polygon: self vertexPoints.





printing


   


printInternalOn: aStream


	aStream print: position





private


   


from: aCollection


	vertices := aCollection collect: [:each | 


		RegionEditorVertex from: each].


	lines := RegionEditorLine linesFromVertices: vertices.


	vertices do: [:vertex | vertex addArea: self].


	lines do: [:line |


		line vertices do: [:vertex | vertex addLine: line].


		line


			addArea: self;


			recalculatePosition].


	self recalculatePosition





instance creation


 


from: oneRegionEditorVertex to: anotherRegionEditorVertex


	^self new 


		from: oneRegionEditorVertex 


		to: anotherRegionEditorVertex


   


linesFromVertices: aCollection


	| stream lines |


	stream := ReadStream on: aCollection, (


		Array with: aCollection first).


	lines := OrderedCollection new: aCollection size + 1.


	aCollection size timesRepeat: [


		lines add: (self from: stream next to: stream peek)].


	^lines











class:									RegionEditorLine


superclass: 						RegionEditorObject 


instanceVariableNames:	vertices areas


poolDictionaries:				OperatingSystemConstants





instance methods





initializing





initialize


	super initialize.


	areas := OrderedCollection new





accessing


 


vertices


	^vertices


  


addArea: anArea


	(areas includes: anArea)


		ifFalse: [areas add: anArea]


  


recalculatePosition


	position := vertices first position 


		leftMostAndHighest: vertices last position


  


replace: aVertex with: anotherVertex


	| index |


	index := vertices indexOf: aVertex.


	vertices at: index put: anotherVertex.


	anotherVertex addLine: self.


	self recalculatePosition


 


areas


	^areas





lowestComponents


	^vertices


  


extent


	^(vertices first position rightMostAndLowest: 


		vertices last position) - position





comparing


  


isSameAs: aRegionEditorLine


	^vertices = aRegionEditorLine vertices or: [


		vertices = aRegionEditorLine vertices reversed]


  


testing





isNearPoint: aPoint


	^self


		doesPoint: aPoint


		collideWithLineFrom: vertices first position


		to: vertices last position


		within: 4


 


isLine


	^true





displaying





displaySelectedWith: aPen


	aPen


		setLineWidth: 3;


		place: vertices first position;


		goto: vertices last position


 


displayWith: aPen


	aPen


		setLineWidth: 1;


		place: vertices first position;


		goto: vertices last position





printing 


  


printInternalOn: aStream


	aStream


		print: vertices first;


		nextPutAll: ' -> ';


		print: vertices last





private





doesPoint: aPoint collideWithLineFrom: startPoint to: endPoint within: threshold


	"If the points coincide, check for the distance 


	between the points only."


	^startPoint = endPoint


		ifTrue: [((aPoint x - startPoint x) squared + 


			(aPoint y - startPoint y) squared) <= 


				threshold squared]


		ifFalse: [self 


			doesPoint: aPoint 


			collideWithArbitraryLineFrom: startPoint 


			to: endPoint 


			within: threshold]


 


doesPoint: p collideWithArbitraryLineFrom: p0 to: p1 within: threshold


	"i represents the projection point of p on the line from p0 to p1"


	| v i |


	v := p1 - p0.


	i := p0 + (v * ((p - p0) dotProduct: v) / (v dotProduct: v)).


	^((i x between: (p0 x min: p1 x) and: (p0 x max: p1 x)) and:


		[i y between: (p0 y min: p1 y) and: (p0 y max: p1 y)]) and:


		[((p x - i x) squared + (p y - i y) squared) <= threshold squared]


   


from: oneVertex to: anotherVertex


	vertices := Array with: oneVertex with: anotherVertex.


	oneVertex addLine: self.


	anotherVertex addLine: self.


	self recalculatePosition





instance creation


  


from: aPoint


	^self new from: aPoint











class:									RegionEditorVertex


superclass: 						RegionEditorObject 


instanceVariableNames:	lines areas


poolDictionaries:				OperatingSystemConstants





instance methods


  


initializing





initialize


	super initialize.


	lines := OrderedCollection new.


	areas := OrderedCollection new





accessing





 position: aPoint


	position := aPoint.


	lines do: [:each |


		each recalculatePosition].


	areas do: [:each |


		each recalculatePosition].





addArea: anArea


	(areas includes: anArea)


		ifFalse: [areas add: anArea]


  


boundingBox


	^super boundingBox expandBy: 2


  


areas


	^areas





addLine: aLine


	(self includesLine: aLine)


		ifFalse: [lines add: aLine]


  


lowestComponents


	^Array with: self


  


extent


	^1@1


 


lines


   ^lines


 


anyArea


	^areas first





testing


   


isNearPoint: aPoint


	^(self boundingBox expandBy: 2) containsPoint: aPoint





isVertex


	^true





includesLine: aLine


	lines do: [:each |


		(each isSameAs: aLine)


			ifTrue: [^true]].


	^false





displaying





displaySelectedWith: aPen


	| oldColor |


	oldColor := aPen backColor.


	aPen


		backColor: aPen foreColor;


		place: self position; circleFilled: 4;


		backColor: oldColor


  


displayWith: aPen


	| box |


	box := (self position extentFromLeftTop: 1 @ 1) 


		expandBy: 1.


	aPen fill: box color: aPen foreColor





printing


   


printInternalOn: aStream


	position printOn: aStream





private


 


from: aPoint


	position := aPoint








class:									DrawingPane


superclass: 						ScrollingPane 


instanceVariableNames:	objects selectedObjects buffer 


poolDictionaries:				OperatingSystemConstants


											VirtualKeyConstants





class methods





events supported


  


constructEventsTriggered


	"Private - answer the set of events that instances of the receiver can trigger."





	"DrawingPane initializeEventsTriggered"





	^super constructEventsTriggered


		add: #controlClickAt:;


		add: #objects:near: ;


		add: #objects:movedBy:;


		add: #objectsDeleted:;


		add: #objectsDisplaySelected:unselected:;


		add: #singleClickSelected:;


		add: #doubleClickSelected:;


		yourself





instance methods





accessing





selectedObjects: aCollection


	selectedObjects := aCollection





selectedObjects


	^selectedObjects





objects


	^objects





pen


	^buffer pen setViewportOrg: super pen getViewportOrg





objects: aCollection


	objects := aCollection





object adding





addObject: anObject


	^self addObject: anObject refresh: true


 


addObjects: aCollection


	^self addObjects: aCollection refresh: true


 


addObject: anObject refresh: aBoolean


	self objects add: anObject.


	aBoolean ifTrue: [self invalidate]


   


addObjects: aCollection refresh: aBoolean


	aCollection do: [:object | self objects add: object].


	aBoolean ifTrue: [self invalidate]





object deleting





deleteObject: anObject


	self deleteObject: anObject refresh: true





deleteObject: anObject refresh: aBoolean


	self objects remove: anObject ifAbsent: [].


	self selectedObjects remove: anObject ifAbsent: [].


	aBoolean ifTrue: [self invalidate]





deleteObjects: aCollection


	self deleteObjects: aCollection refresh: true





deleteObjects: aCollection refresh: aBoolean


	aCollection do: [:object |


		self objects remove: object ifAbsent: [].


		self selectedObjects remove: object ifAbsent: []].


	aBoolean ifTrue: [self invalidate]





object selecting


 


selectObject: anObject


	self selectObject: anObject refresh: true





selectObject: anObject refresh: aBoolean


	self selectedObjects: (OrderedCollection with: anObject).


	aBoolean ifTrue: [self invalidate]





selectObjects: aCollection


	self selectObjects: aCollection refresh: true





selectObjects: aCollection refresh: aBoolean


	self selectedObjects: aCollection.


	aBoolean ifTrue: [self invalidate]





private-nitializing


   


defaultCursor


	^CursorManager crossHair


  


initialize


	super initialize.


	self objects: OrderedCollection new.


	self selectedObjects: OrderedCollection new.





private-buffer





destroyBuffer


	buffer isNil ifFalse: [buffer release]





createBuffer


	self destroyBuffer.


	buffer := Bitmap screenExtent: 


		self visibleRectangle extent.


	self pen


		foreColor: self foreColor;


		backColor: self backColor





private-displaying





displayIn: aRectangle


	super displayIn: aRectangle.


	self display





displayBuffer


	super pen


		copyBitmap: buffer


		from: self visibleRectangle


		to: self visibleRectangle


 


displayObjects


	| unselected |


	unselected := self objects reject: [:object |


		self selectedObjects includes: object].


	self triggerEvent: #objectsDisplaySelected:unselected:


		with: self selectedObjects with: unselected





display


	self pen


		fill: self visibleRectangle


		color: self backColor.


	self


		displayObjects;


		displayBuffer





private-validating





validate


	super validate.


	self createBuffer





private-resizing





sizeChanged: anExtent


	super sizeChanged: anExtent.


	self createBuffer





private-selecting





moveFrom: aPoint


	"Inform owner of each move."


	| start |


	start := aPoint. self invalidate.


	self whileButton1DownDo: [:location |


		self triggerEvent: #objects:movedBy: 


			with: self selectedObjects with: location - start.


		start := self mouseLocation.


		self display].


   


normalSelect: aPoint


	"Selecting a preselected object begins a move; otherwise, a new list of selections is computed. Inform owner afterwards."


	| list |


	list := self triggerEvent: #objects:near: 


		with: self selectedObjects with: aPoint.


	list isEmpty ifFalse: [^self moveFrom: aPoint].


	list := self triggerEvent: #objects:near: 


		with: self objects with: aPoint.


	self selectObjects: list.


	self triggerEvent: #singleClickSelected: with: list.


	self invalidate.


	self moveFrom: aPoint





controlSelect: aPoint


	"User-defined activity required. Informs owner."


	self triggerEvent: #controlClickAt: with: aPoint.


	self invalidate


  


shiftSelect: aPoint


	"Automatically adds/removes from selected objects. Informs owner afterwards."


	| list |


	list := self triggerEvent: #objects:near: 


		with: self objects with: aPoint.


	list do: [:object |


			(self selectedObjects includes: object)


				ifTrue: [self selectedObjects remove: object]


				ifFalse: [self selectedObjects add: object]].


	self triggerEvent: #singleClickSelected: 


		with: self selectedObjects.


	self invalidate





private-closing





close


	super close.


	buffer release





private-window events





button1DoubleClick: aPoint


	"Handle 3 modifier possibilities in canvas coordinates."


	super button1DoubleClick: aPoint.


	self triggerEvent: #doubleClickSelected: 


		with: self selectedObjects.


	self invalidate





keyboardInput: aKeyboardInputEvent


	"Private - keyboard input was received.  Process backspace, tab, carriage return."


	| virtualKey |


	( virtualKey := aKeyboardInputEvent virtualKey ) notNil 	ifTrue: [


			(virtualKey == BackspaceKey or: [


				virtualKey == DeleteKey])


					ifTrue: [self deleteSelectedObjects]].





button1Down: aPoint


	"Handle 3 modifier possibilities in canvas coordinates."


	super button1Down: aPoint.


	self isShiftKeyDown ifTrue: [^self shiftSelect: aPoint].


	self isControlKeyDown ifTrue: [


		^self controlSelect: aPoint].


	^self normalSelect: aPoint


 


deleteSelectedObjects


	| candidates |


	candidates := self selectedObjects shallowCopy.


	self deleteObjects: candidates refresh: false.


	self triggerEvent: #objectsDeleted: with: candidates; invalidate


 


button2Up: aPoint


	super button2Up: aPoint.


	self triggerEvent: #needsMenu: with: self selectedObjects

